Use of Fourier serie to solve parabolic PDE

```
> restart:
```

Problem: solve the PDE

```
> PDE := diff(T(t,x),t) = mu * diff(T(t,x),x,x) ;
```

$$PDE := \frac{\partial}{\partial t}\, T(t,x) = \mu \left(\frac{\partial^2}{\partial x^2}\, T(t,x)\right) \quad \textbf{(1)}$$

With BC:

```
> BC0  := T(0,x)  = T0(x);
  BC_L := T(t,0)  = 0;
  BC_R := T(t,Pi) = 0;
```

$$BC0 := T(0,x) = T0(x)$$

$$BC\_L := T(t,0) = 0$$

$$BC\_R := T(t,\pi) = 0 \quad \textbf{(2)}$$

Obtain a particular solution as f(t)*g(x)

```
> PFUN := f(t)*g(x) ;
```

$$PFUN := f(t)\, g(x) \quad \textbf{(3)}$$

```
> subs(T(t,x)=f(t)*g(x),PDE) ; TWO_ODE := expand(%)/(f(t)*g(x)) ;
```

$$\frac{\partial}{\partial t}\,(f(t)\, g(x)) = \mu \left(\frac{\partial^2}{\partial x^2}\,(f(t)\, g(x))\right)$$

$$TWO\_ODE := \frac{\frac{d}{dt} f(t)}{f(t)} = \frac{\mu \left(\frac{d^2}{dx^2}\, g(x)\right)}{g(x)} \quad \textbf{(4)}$$

```
> ODE1 := lhs(TWO_ODE) = -mu*k^2 ;
  ODE2 := rhs(TWO_ODE) = -mu*k^2 ;
```

$$ODE1 := \frac{\frac{d}{dt} f(t)}{f(t)} = -\mu\, k^2$$

$$ODE2 := \frac{\mu \left(\frac{d^2}{dx^2}\, g(x)\right)}{g(x)} = -\mu\, k^2 \quad \textbf{(5)}$$

```
> SOL1 := dsolve( {ODE1,f(0)=f0}  ) ;
```

$$SOL1 := f(t) = f0\, e^{-\mu k^2 t} \quad \textbf{(6)}$$

```
> SOL2 := dsolve( ODE2 ) ;
```

$$SOL2 := g(x) = \_C1 \sin(k\, x) + \_C2 \cos(k\, x) \quad \textbf{(7)}$$

The particular solution is of the form

```
> PSOLFOUND := subs(SOL2,subs(SOL1,f(t)*g(x))) ;
```

$$PSOLFOUND := f0\, e^{-\mu k^2 t}\,(\_C1 \sin(k\, x) + \_C2 \cos(k\, x)) \quad \textbf{(8)}$$

There are other 2 particular solutions

```
> subs(T(t,x)=a0/2+b*x,PDE) ; expand(%);
```

$$\frac{\partial}{\partial t}\left(\frac{1}{2}\,a0+b\,x\right)=\mu\left(\frac{\partial^2}{\partial x^2}\left(\frac{1}{2}\,a0+b\,x\right)\right)$$

$$0=0 \qquad (9)$$

The general solution can be written as a summation:

```
> TGENERAL := a0/2+b*x+Sum( exp(-mu*k^2*t)*
                  (a[k]*cos(k*x)+b[k]*sin(k*x)),k=1..infinity) ;
```

$$TGENERAL := \frac{1}{2}\,a0+b\,x+\sum_{k=1}^{\infty}\mathrm{e}^{-\mu\,k^2\,t}\left(a_k\cos(k\,x)+b_k\sin(k\,x)\right) \qquad (10)$$

Initial condition

```
> subs(t=0,TGENERAL);
```

$$\frac{1}{2}\,a0+b\,x+\sum_{k=1}^{\infty}\mathrm{e}^{0}\left(a_k\cos(k\,x)+b_k\sin(k\,x)\right) \qquad (11)$$

Left border

```
> subs(x=0,TGENERAL); simplify(%);
```

$$\frac{1}{2}\,a0+\sum_{k=1}^{\infty}\mathrm{e}^{-\mu\,k^2\,t}\left(a_k\cos(0)+b_k\sin(0)\right)$$

$$\frac{1}{2}\,a0+\sum_{k=1}^{\infty}\mathrm{e}^{-\mu\,k^2\,t}\,a_k \qquad (12)$$

Right border

```
> subs(x=Pi,TGENERAL); simplify(%);
```

$$\frac{1}{2}\,a0+b\,\pi+\sum_{k=1}^{\infty}\mathrm{e}^{-\mu\,k^2\,t}\left(a_k\cos(k\,\pi)+b_k\sin(k\,\pi)\right)$$

$$\frac{1}{2}\,a0+b\,\pi+\sum_{k=1}^{\infty}\mathrm{e}^{-\mu\,k^2\,t}\,a_k\,(-1)^k \qquad (13)$$